Supplementary data
Fitting dispersal kernels for winged and wingless achenes of Pallenis spinosa using a Bayesian approach.

Model code is written in OpenBUGS language and is based on Snäll et al (2007). For line source experiments (years 1 and 3), marginalized versions to one dimension of the three tested dispersal kernels (probability density functions) were used, whereas for the point source experiment (year 2) we fitted two-dimensional kernels (Table 1). Hierarchical models were constructed for year 1 and year 2 experiments, in which three replicates of the experiment and four circular sectors, respectively, were established.
Non-informative prior distributions were specified, except for the case of parameter c of the two-dimensional 2Dt curve (year 2 experiment). To be able to fit the model in this case, the prior distribution was set to encompass essentially the interval 0.5-1, following Clark et al. (1999). The models were fitted by running four chains after an initial burn-in of 10000 iterations and then retaining every 100th iteration for a total of 1.5 million iterations. 
Year 1 experiment: line source, hierarchical model including three replicates
Model code (text after the sign # is not compiled, log refers to natural logarithm)
model{

for(i in 1:N){

Y[i] ~ dpois(mu[i])

mu.temp[i] <- Lk[t[i]] + log(numtr[i])+Lc[t[i]]-(Lal[t[i]]+loggam(1/exp(Lc[t[i]])))-exp(exp(Lc[t[i]])*(log(x[i])-Lal[t[i]])) #GEC (generalized exponential curve)
#mu.temp[i] <- Lk[t[i]] + log(numtr[i])-Lal[t[i]]-exp(log(x[i])-Lal[t[i]]) #EC (exponential curve)
#mu.temp[i] <- Lk[t[i]] + log(numtr[i])+loggam((exp(Lc[t[i]])+1)/2) - loggam(exp(Lc[t[i]])/2)- Lal[t[i]]-log(sqrt(exp(Lc[t[i]])*3.1416))-((exp(Lc[t[i]])+1)/2)*log(1+exp(2*(log(x[i])-Lal[t[i]])-Lc[t[i]])) #2Dt curve
Lmu[i] ~ dnorm(mu.temp[i],tau.mu) # Overdispersion, tau=1/(sigma^2)

log(mu[i]) <- Lmu[i]

}

# Hierarchical structure

for(j in 1:T){

Lal[j] ~ dnorm(Lal.mu, Lal.tau)

Lc[j] ~ dnorm(Lc.mu, Lc.tau)

}

# Prior distributions

# First term: mean; second term: precision=1/(standard deviation^2)

# The “L” before the parameter name refers to “log of”

tau.mu ~ dgamma(0.001,0.001)

Lk[1] ~ dnorm(0,0.0001)

Lk[2] ~ dnorm(0,0.0001)

Lk[3] ~ dnorm(0,0.0001)

Lal.mu ~ dnorm(0,0.1)

Lal.tau ~ dgamma(0.001,0.001)

Lc.mu ~ dnorm(0,0.1)

Lc.tau ~ dgamma(0.001,0.001)

}

Data:

list( N=24, T=3,

t=c(1,1,1,1,1,1,1,1,2,2,2,2,2,2,2,2,3,3,3,3,3,3,3,3), #replicate
x=c(0.25,0.5,0.75,1,1.5,2,3,5,0.25,0.5,0.75,1,1.5,2,3,5,0.25,0.5,0.75,1,1.5,2,3,5), #distance (m)
numtr=c(6,6,6,6,9,9,9,9,6,6,6,6,9,9,9,9,6,6,6,6,9,9,9,9), #number of seed traps
#Y=c(280,215,146,100,63,35,27,2,201,175,157,128,167,115,51,5,436,185,160,115,138,93,39,5)) #winged achenes
Y=c(783,406,209,142,100,64,21,5,1434,741,359,233,231,143,55,12,1620,633,325,233,183,105,34,6)) #wingless achenes
GEC
Initial values, winged achenes:

list( tau.mu = 33,Lk=c(6.5,6.0,6.8),Lal.mu=-0.3,Lal.tau=71.0,Lc.mu=-0.25,Lc.tau=150)

list( tau.mu = 35,Lk=c(6.1,6.3,6.9),Lal.mu=-0.25,Lal.tau=70.0,Lc.mu=-0.18,Lc.tau=170)

list( tau.mu = 31,Lk=c(6.8,6.2,6.5),Lal.mu=-0.22,Lal.tau=65.0,Lc.mu=-0.3,Lc.tau=180)

list( tau.mu = 27,Lk=c(6.6,6.7,6.8),Lal.mu=-0.26,Lal.tau=68.0,Lc.mu=-0.15,Lc.tau=160)

Initial values, wingless achenes:

list( tau.mu = 58,Lk=c(8.5,8.0,8.8),Lal.mu=-4.0,Lal.tau=160.0,Lc.mu=-1.0,Lc.tau=450)

list( tau.mu = 55,Lk=c(8.1,8.3,8.9),Lal.mu=-4.4,Lal.tau=150.0,Lc.mu=-1.2,Lc.tau=470)

list( tau.mu = 60,Lk=c(8.8,8.2,8.5),Lal.mu=-4.2,Lal.tau=155.0,Lc.mu=-1.3,Lc.tau=480)

list( tau.mu = 62,Lk=c(8.3,8.3,8.4),Lal.mu=-4.9,Lal.tau=156.0,Lc.mu=-1.28,Lc.tau=490)

EC

Initial values, winged achenes:

list( tau.mu = 20,Lk=c(6.5,6.0,6.8),Lal.mu=-0.3,Lal.tau=110.0)

list( tau.mu = 21,Lk=c(6.1,6.3,6.9),Lal.mu=-0.25,Lal.tau=105.0)

list( tau.mu = 22,Lk=c(6.8,6.2,6.5),Lal.mu=-0.22,Lal.tau=100.0)

list( tau.mu = 23,Lk=c(6.6,6.7,6.8),Lal.mu=-0.26,Lal.tau=102.0)

Initial values, wingless achenes:

list( tau.mu = 6,Lk=c(7.5,7.0,7.8),Lal.mu=-0.2,Lal.tau=310.0)

list( tau.mu = 6.5,Lk=c(7.1,7.3,7.9),Lal.mu=-0.4,Lal.tau=280.0)

list( tau.mu = 6.8,Lk=c(7.8,7.2,7.5),Lal.mu=-0.3,Lal.tau=300.0)

list( tau.mu = 6.7,Lk=c(7.3,7.3,7.4),Lal.mu=-0.9,Lal.tau=290.0)

2Dt
Initial values, winged achenes:

list( tau.mu = 23,Lk=c(7.5,7.0,7.8),Lal.mu=-0.3,Lal.tau=51.0,Lc.mu=0.9,Lc.tau=132)

list( tau.mu = 21,Lk=c(7.1,7.3,7.9),Lal.mu=-0.12,Lal.tau=50.0,Lc.mu=0.8,Lc.tau=127)

list( tau.mu = 20,Lk=c(7.8,7.2,7.5),Lal.mu=-0.2,Lal.tau=55.0,Lc.mu=1.0,Lc.tau=125)

list( tau.mu = 17,Lk=c(7.6,7.7,7.8),Lal.mu=-0.1,Lal.tau=48.0,Lc.mu=0.5,Lc.tau=120)

Initial values, wingless achenes:

list( tau.mu = 68,Lk=c(8.5,8.0,8.8),Lal.mu=-1.0,Lal.tau=260.0,Lc.mu=-1.0,Lc.tau=250)

list( tau.mu = 75,Lk=c(8.1,8.3,8.9),Lal.mu=-1.4,Lal.tau=250.0,Lc.mu=-0.1,Lc.tau=270)

list( tau.mu = 70,Lk=c(8.8,8.2,8.5),Lal.mu=-1.2,Lal.tau=255.0,Lc.mu=0.3,Lc.tau=220)

list( tau.mu = 72,Lk=c(8.3,8.3,8.4),Lal.mu=-0.9,Lal.tau=256.0,Lc.mu=-0.25,Lc.tau=230)

Year 2 experiment: point source, hierarchical model including four circular sectors
model{

for(i in 1:N){

Y[i] ~ dpois(mu[i])

mu.temp[i] <- Lk[d[i]]+log(numtr[i])+Lc[d[i]]-(log(2*3.1416)+2*Lal[d[i]]+loggam(2 / exp(Lc[d[i]])))-exp(exp(Lc[d[i]])*(log(x[i])-Lal[d[i]])) #GEC

#mu.temp[i] <- Lk[d[i]]+log(numtr[i])-(log(2*3.1416)+2*Lal[d[i]])-exp(log(x[i])-Lal[d[i]]) #EC

#mu.temp[i] <- Lk[d[i]]+log(numtr[i])+Lc[d[i]]-log(3.1416)-Lal[d[i]]-(exp(Lc[d[i]])+1)*log(1+pow(x[i],2)/exp(Lal[d[i]])) #2Dt

Lmu[i] ~ dnorm(mu.temp[i],tau.mu) # Overdispersion,tau=1/(sigma^2)

log(mu[i]) <- Lmu[i]

}

# Hierarchical structure

# Effect of direction on parameters

 for(p in 1:D){

   Lk[p] ~ dnorm(Lk.mu, Lk.tau)

   Lal[p] ~ dnorm(Lal.mu, Lal.tau)

   Lc[p] ~ dnorm(Lc.mu, Lc.tau)

}

# Prior distributions

# First term: mean; second term: precision=1/(standard deviation^2)

# The “L” before the parameter name refers to “log of”

tau.mu ~ dgamma(0.001,0.001)

Lk.mu ~ dnorm(0,0.0001)

Lal.mu ~ dnorm(0,0.1) 

Lc.mu ~ dnorm(0,0.1) #only for GEC

#Lc.mu ~ dnorm(-0.288,3) #only for 2Dt curve(parameter c values essentially in the range 0.5-1 (Clark et al., 1999)

Lk.tau ~ dgamma(0.001,0.001)

Lal.tau ~ dgamma(0.001,0.001)

Lc.tau ~ dgamma(0.001,0.001)

}

Data:

list( N=25, D=4,

d=c(1,1,1,1,1,1,1,1,1,1,2,2,2,2,2,3,3,3,3,3,4,4,4,4,4), #circular sectors (1, N-NE; 2, E-SE; 3: S-SW; 4, W-NW)
x=c(0.4,0.8,1.2,2,3,5,10,20,30,40,0.4,0.8,1.2,2,3,0.4,0.8,1.2,2,3,0.4,0.8,1.2,2,3), #distance (m)
numtr=c(1,2,2,4,4,4,10,16,24,32,1,2,2,4,4,1,2,2,4,4,1,2,2,4,4), #number of seed traps
#Y=c(88,87,45,24,13,1,0,0,0,0,43,29,36,15,1,35,27,21,4,5,29,25,9,6,2)) #winged achenes
Y=c(123,106,40,41,9,8,0,0,0,0,213,102,48,36,10,307,227,136,97,25,227,153,84,49,18)) #wingless achenes
Initial values:

GEC,winged achenes
list( tau.mu=51.0,Lk.mu=13,Lal.mu=-2.45,Lc.mu=-0.53,Lk.tau=6,Lal.tau=245,Lc.tau=480)

list( tau.mu=55.0,Lk.mu=14,Lal.mu=-2.55,Lc.mu=-0.525,Lk.tau=5.5,Lal.tau=238,Lc.tau=550)

list( tau.mu=50,Lk.mu=12,Lal.mu=-2.47,Lc.mu=-0.52,Lk.tau=4.5,Lal.tau=250,Lc.tau=535)

list( tau.mu=56,Lk.mu=9.2,Lal.mu=-2.39,Lc.mu=-0.54,Lk.tau=4,Lal.tau=240,Lc.tau=557)

GEC,wingless achenes

list( tau.mu=245,Lk.mu=11.1,Lal.mu=-6.0,Lc.mu=-1.12,Lk.tau=4.55,Lal.tau=220,Lc.tau=642)

list( tau.mu=241,Lk.mu=10.6,Lal.mu=-6.1,Lc.mu=-1.1,Lk.tau=6.3,Lal.tau=195,Lc.tau=695)

list( tau.mu=242,Lk.mu=11.0,Lal.mu=-5.9,Lc.mu=-1.15,Lk.tau=4.8,Lal.tau=215,Lc.tau=690)

list( tau.mu=230,Lk.mu=10.9,Lal.mu=-5.95,Lc.mu=-1.05,Lk.tau=4.9,Lal.tau=187,Lc.tau=655)

EC,winged achenes

list( tau.mu=65.0,Lk.mu=9.4,Lal.mu=-0.38,Lk.tau=3.5,Lal.tau=320)

list( tau.mu=57.0,Lk.mu=9.3,Lal.mu=-0.35,Lk.tau=3.0,Lal.tau=325)

list( tau.mu=70,Lk.mu=9.5,Lal.mu=-0.4,Lk.tau=4.5,Lal.tau=330)

list( tau.mu=60,Lk.mu=9.2,Lal.mu=-0.45,Lk.tau=4,Lal.tau=288)

EC,wingless achenes

list( tau.mu=6.0,Lk.mu=10.1,Lal.mu=-0.5,Lk.tau=4.55,Lal.tau=220)

list( tau.mu=7.0,Lk.mu=10.6,Lal.mu=-0.3,Lk.tau=4.3,Lal.tau=225)

list( tau.mu=8.0,Lk.mu=11.0,Lal.mu=-0.35,Lk.tau=4.8,Lal.tau=215)

list( tau.mu=7.5,Lk.mu=10.9,Lal.mu=-0.25,Lk.tau=4.9,Lal.tau=217)

2Dt,winged achenes
list(tau.mu=92,Lk.mu=9.4,Lal.mu=-0.23,Lc.mu=-1.0,Lk.tau=5.5,Lc.tau=246,Lal.tau=212)

list(tau.mu=80,Lk.mu=9.2,Lal.mu=-0.2,Lc.mu=-0.5,Lk.tau=6,Lc.tau=235,Lal.tau=100)

list(tau.mu=86,Lk.mu=9.8,Lal.mu=-0.35,Lc.mu=-0.4,Lk.tau=5,Lc.tau=250,Lal.tau=225)

list( tau.mu=90,Lk.mu=9.5,Lal.mu=-0.33,Lc.mu=-0.6,Lk.tau=6.5,Lc.tau=230,Lal.tau=235)
2Dt,wingless achenes
list( tau.mu=101,Lk.mu=11.4,Lal.mu=-2.05,Lc.mu=-1.0,Lk.tau=11,Lal.tau=120,Lc.tau=210)

list( tau.mu=100,Lk.mu=11.6,Lal.mu=-2.1,Lc.mu=-1.2,Lk.tau=12,Lal.tau=110,Lc.tau=200)

list( tau.mu=102,Lk.mu=11.2,Lal.mu=-1.9,Lc.mu=-1.3,Lk.tau=13,Lal.tau=117,Lc.tau=220)

list( tau.mu=109,Lk.mu=11.1,Lal.mu=-2.3,Lc.mu=-1.35,Lk.tau=10.5,Lal.tau=123,Lc.tau=215)

Year 3 experiment: line source, non-hierarchical model

model{

for(i in 1:N){

Y[i] ~ dpois(mu[i])

#mu.temp[i] <- Lk + log(numtr[i])+ Lc-(Lal+loggam(1/exp(Lc)))-exp(exp(Lc)*(log(x[i])-Lal)) #GEC

#mu.temp[i] <- Lk + log(numtr[i])-Lal-exp(log(x[i])-Lal) #EC

mu.temp[i] <- Lk + log(numtr[i])+ loggam((exp(Lc)+1)/2) - loggam(exp(Lc)/2)- Lal-log(sqrt(exp(Lc)*3.1416))-((exp(Lc)+1)/2)*log(1+exp(2*(log(x[i])-Lal)-Lc)) #2Dt

Lmu[i] ~ dnorm(mu.temp[i],tau.mu) # Overdispersion,tau=1/(sigma^2)

log(mu[i]) <- Lmu[i]

}

# Prior distributions

# First term: mean; second term: precision=1/(standard deviation^2)

# The “L” before the parameter name refers to “log of”

tau.mu ~ dgamma(0.001,0.001)

Lk ~ dnorm(0,0.0001)

Lal ~ dnorm(0,0.1)

Lc ~ dnorm(0,0.1)

}

Data:

list( N=13, 

x=c(0.5,1,1.5,2,2.5,3,3.5,4,5,6,8,10,12), #distance (m)

numtr=c(4,5,6,6,7,8,9,10,11,12,13,14,15), #number of seed traps

#Y=c(54,44,25,21,11,4,4,5,3,1,0,0,0)) #winged achenes

Y=c(305,246,202,106,102,74,60,38,26,8,1,0,1)) #wingless achenes

Initial values:

GEC,winged achenes

list( tau.mu = 260,Lk=7.1,Lal=-0.8,Lc=-0.25)

list( tau.mu = 250,Lk=7.5,Lal=-0.75,Lc=-0.20)

list( tau.mu = 255,Lk=7.9,Lal=-0.85,Lc=-0.3)

list( tau.mu = 270,Lk=7.9,Lal=-0.78,Lc=-0.28)

GEC,wingless achenes

list( tau.mu = 500,Lk=8.8,Lal=-0.5,Lc=-0.4)

list( tau.mu = 495,Lk=8.9,Lal=-0.6,Lc=-0.5)

list( tau.mu = 470,Lk=8.7,Lal=-0.4,Lc=-0.7)

list( tau.mu = 490,Lk=9.6,Lal=-0.3,Lc=-0.3)

EC,winged achenes

list( tau.mu = 310,Lk=7.1,Lal=-0.05)

list( tau.mu = 315,Lk=7.5,Lal=-0.1)

list( tau.mu = 295,Lk=7.9,Lal=-0.15)

list( tau.mu = 290,Lk=7.9,Lal=-0.08)

EC,wingless achenes

list( tau.mu = 500,Lk=8.8,Lal=-0.5)

list( tau.mu = 495,Lk=8.9,Lal=-0.6)

list( tau.mu = 470,Lk=8.7,Lal=-0.4)

list( tau.mu = 490,Lk=9.6,Lal=-0.3)

2Dt,winged achenes

list( tau.mu = 280,Lk=7.1,Lal=-0.05,Lc=1.25)

list( tau.mu = 290,Lk=7.5,Lal=-0.06,Lc=1.20)

list( tau.mu = 295,Lk=7.9,Lal=-0.07,Lc=1.3)

list( tau.mu = 285,Lk=7.9,Lal=-0.08,Lc=1.28)

2Dt,wingless achenes

list( tau.mu = 170,Lk=9.8,Lal=0.25,Lc=1.3)

list( tau.mu = 185,Lk=9.9,Lal=0.28,Lc=1.4)

list( tau.mu = 190,Lk=9.7,Lal=0.3,Lc=1.5)

list( tau.mu = 160,Lk=9.6,Lal=0.22,Lc=1.35)
